

Folgen, Konvergenz und Häufungspunkte

[Aufgaben: 8

[> **restart**;

Grundlegende Definitionen

MATH: Folgen, genauer unendliche X -wertige Folgen, sind Abbildungen von der Menge

$$\mathbb{N} := \{1, 2, 3, \dots\}$$

der natürlichen Zahlen in eine Menge X . Sie werden oft geschrieben als (a_i) ,

womit gemeint ist

$$a: \mathbb{N} \rightarrow X: i \mapsto a_i$$

Für uns sind zunächst reellwertige Folgen interessant, d. h. $X := \mathbb{R}$ ist die Menge der reellen Zahlen. Die Menge der X -wertigen Folgen bezeichnen wir mit $X^{\mathbb{N}}$.

MAPLE: Das MAPLE-Kommando **seq** lässt nur endliche Folgen zu. Deshalb werden wir Folgen als Abbildungen in MAPLE definieren, wobei uns klar sein muss, dass der Definitionsbereich \mathbb{N} ist. (MAPLE weiß das nicht! Man könnte es MAPLE mitteilen, bekommt aber dann bei gewissen Operationen Schwierigkeiten.)

> **a:=i->i^2;**

$$a := i \rightarrow i^2 \quad (1.1.1)$$

> **a(12);**

$$144 \quad (1.1.2)$$

> **map(a,[\$1..20]);**

[1, 4, 9, 16, 25, 36, 49, 64, 81, 100, 121, 144, 169, 196, 225, 256, 289, 324, 361, 400] (1.1.3)

Da diese Folge natürliche Zahlen als Werte annimmt, kann man sie mit sich selber komponieren (im Sinne von zweimal hintereinander anwenden) und bekommt wieder eine Folge:

> **(a@a)(2);**

$$16 \quad (1.1.4)$$

> **map(a@a,[\$1..10]);**

[1, 16, 81, 256, 625, 1296, 2401, 4096, 6561, 10000] (1.1.5)

Man sieht, dass die Folge der 4. Potenzen aus der Folge der Quadrate als Teilfolge konstruiert werden kann, wobei man sich bei einer Teilfolge vorstellt, dass Glieder der ursprünglichen Folge weggelassen wurden.

MATH: Eine reellwertige Folge a heißt streng monoton steigend, falls

$$a_{i+1} > a_i$$

für alle $i \in \mathbb{N}$ gilt. (Entsprechend definiert man die Begriffe "monoton steigend", "streng monoton fallend" und "monoton fallend", indem man $>$ durch \geq , $<$

bzw. \leq ersetzt.)

MATH: Eine Teilfolge einer Folge f ist die Komposition $f \circ g$ von f mit einer streng monoton steigenden Folge g mit Wertebereich \mathbb{N} .

> **g:=i->2*i+3;**

$$g := i \rightarrow 2i + 3 \quad (1.1.6)$$

> **map(a@g,[\$1..20]);**

$$[25, 49, 81, 121, 169, 225, 289, 361, 441, 529, 625, 729, 841, 961, 1089, 1225, 1369, 1521, 1681, 1849] \quad (1.1.7)$$

MATH: Für \mathbb{R}^* -wertige Folgen, also reellwertige Folgen, die niemals den Wert Null annehmen, kann man die Quotientenfolge einführen, welche wie folgt definiert ist:

$$QuO(f) : \mathbb{N} \rightarrow \mathbb{R}^* : n \mapsto \frac{f_{n+1}}{f_n}$$

Diese ist in unserer Datenstruktur gegeben durch:

> **QuO:=proc(f::procedure)**
 unapply(simplify(f(n+1)/f(n)),n);
 end proc;

> **QuO(g);**

$$n \rightarrow \frac{2n+5}{2n+3} \quad (1.1.8)$$

MATH: Eine Folge a mit positiven Werten ist genau dann streng monoton steigend, wenn $QuO(a)$ nur Werte > 1 annimmt. (Entsprechendes gilt für die Begriffe "monoton steigend", "streng monoton fallend" und "monoton fallend", indem man $>$ durch \geq , $<$ bzw. \leq ersetzt.)

BEISPIEL: $n \mapsto \left(1 + \frac{1}{n}\right)^n$ ist eine streng monoton steigende Folge:

> **a:=n->(1+1/n)^n;**

$$a := n \rightarrow \left(1 + \frac{1}{n}\right)^n \quad (1.1.9)$$

Dies ist sicher eine positive Folge, so dass wir mit der Quotientenfolge versuchen können, die Monotonie zu entscheiden:

a ist genau dann streng monoton steigend, wenn

> **QuO(a);**

$$n \rightarrow \frac{\left(\frac{n+2}{n+1}\right)^n (n+2) \left(\frac{n+1}{n}\right)^{-n}}{n+1} \quad (1.1.10)$$

größer als 1 ist.

> **QuO(a)(n)=(1-1/(n+1)^2)^n*(1+1/(n+1));**

$$\frac{\left(\frac{n+2}{n+1}\right)^n (n+2) \left(\frac{n+1}{n}\right)^{-n}}{n+1} = \left(1 - \frac{1}{(n+1)^2}\right)^n \left(1 + \frac{1}{n+1}\right) \quad (1.1.11)$$

> **simplify(%)**;

$$\frac{\left(\frac{n+2}{n+1}\right)^n (n+2) \left(\frac{n+1}{n}\right)^{-n}}{n+1} = \frac{\left(\frac{n(n+2)}{(n+1)^2}\right)^n (n+2)}{n+1} \quad (1.1.12)$$

Mithilfe der Bernoulli-Ungleichung bekommt man:

$$\frac{a_n}{a_{n-1}} = \left(1 - \frac{1}{(n+1)^2}\right)^n \cdot \left(1 + \frac{1}{n+1}\right) \geq \left(1 - \frac{n}{(n+1)^2}\right) \cdot \left(1 + \frac{1}{n+1}\right).$$

> **aa:=simplify((1-n/(n+1)^2)*(1+1/(n+1)));**

$$aa := \frac{(n^2 + n + 1)(n + 2)}{(n + 1)^3} \quad (1.1.13)$$

> **expand(numer(aa));**
expand(denom(aa));

$$\frac{n^3 + 3n^2 + 3n + 2}{n^3 + 3n^2 + 3n + 1} \quad (1.1.14)$$

also ≥ 1 . Damit ist a_n monoton steigend.

ÜBUNG [01]:

Zeige, dass die Folge b monoton fallend ist.

> **b:=n->(1+1/n)^(n+1);**

$$b := n \rightarrow \left(1 + \frac{1}{n}\right)^{n+1} \quad (1.1.15)$$

▼ Konvergenz von Folgen

MATH: Eine reellwertige Folge $a = (a_i) : \mathbb{N} \rightarrow \mathbb{R}$ heißt **konvergent** gegen den

Grenzwert g , falls gilt:

Zu jedem $\varepsilon \in \mathbb{R}_{>0}$ existiert ein $N_0 = N_0(\varepsilon) \in \mathbb{N}$ mit $|a_i - g| < \varepsilon$ für alle $i \in \mathbb{N}$ mit $i > N_0$.

Folgen mit Grenzwert 0 nennt man **Nullfolgen**.

MAPLE: Um ein erstes Gefühl für Konvergenz zu bekommen, geben wir eine Genauigkeit vor und schauen, ab welchem

N_0 die Folge innerhalb der gegebenen

Genauigkeit konstant ist:

```
> f:=i->1/i;
```

$$f:=i \rightarrow \frac{1}{i}$$

(1.2.1)

Z.B. für die obige Folge a_i und Genauigkeit 10^{-2} .

```
> for j from 195 to 205 do
  round(10^2*f(j))/10^2
end do;
j:='j':
```

$$\frac{1}{100}$$

$$\frac{1}{100}$$

$$\frac{1}{100}$$

$$\frac{1}{100}$$

$$\frac{1}{100}$$

$$\frac{1}{100}$$

$$0$$

$$0$$

$$0$$

$$0$$

$$0$$

(1.2.2)

MATH: Natürlich beweist dieses Experiment noch lange nicht, dass die Folge auch konvergiert. Aber es gibt Sätze, mit deren Hilfe man Konvergenz beweisen kann: Z. B.:

Monoton steigende (fallende) Folgen, die beschränkt sind, sind konvergent.

(Beachte: Den Grenzwert hat man damit noch lange nicht!)

Eine Folge a heißt **beschränkt**, falls ein $K \in \mathbb{R}$ existiert mit

$$|a_n| < K$$

für alle $n \in \mathbb{N}$.

MATH: Übrigens ist nicht jede Folge, die langsam monoton steigt, konvergent.

Hier ist ein Beispiel, welches gleichzeitig zu einem vorsichtigen Umgang mit

MAPLE mahnt.

```
> h:=n->sum(1/i,i=1..n);
```

(1 2 3)

$$h := n \rightarrow \sum_{i=1}^n \frac{1}{i} \quad (1.2.3)$$

> h(n);

$$\Psi(n+1) + \gamma \quad (1.2.4)$$

Solche Antworten sollten dich nicht aus dem seelischen Gleichgewicht bringen: In MAPLE ist mathematisches Wissen berücksichtigt, welches du im Augenblick noch nicht hast, aber sicherlich im Laufe der Zeit erwerben wirst. Fasse MAPLEs Antwort als ein Signal auf, dass MAPLE sich wohl fühlt.

> for k from 1100 to 1110 do evalf(h(k)) end do;

7.580735600

7.581643866

7.582551307

7.583457925

7.584363722

7.585268699

7.586172859

7.587076201

7.587978728

7.588880441

7.589781342

(1.2.5)

Die Monotonie unserer Folge ist leicht zu erkennen. Aber leider ist sie nicht nach oben beschränkt, denn

> n := 'n':

> Sum(1/m, m=2^(n-1)+1..2^n);

$$\sum_{m=2^{n-1}+1}^{2^n} \frac{1}{m} \quad (1.2.6)$$

ist sicherlich größer oder gleich

> Sum(1/2^n, m=2^(n-1)+1..2^n) = sum(1/2^n, m=2^(n-1)+1..2^n);

$$\sum_{m=2^{n-1}+1}^{2^n} \frac{1}{2^n} = \frac{2^n - 2^{n-1}}{2^n} \quad (1.2.7)$$

> simplify(rhs(%));

$$\frac{1}{2}$$

(1.2.8)

ÜBUNG [02]:

- 1) Warum ist mit der letzten Abschätzung die Divergenz der Folge h bewiesen?
- 2) Warum sollte man bei folgender Lotterie nicht teilnehmen?
- 3) Wie lange reichen 10 Euro, um den Gewinn bei der Loterie auszuzahlen?

THE MATHEMATICIAN'S LOTTERY

PRIZES OF €750 TRILLION*

ALL LOTTERY PROFITS ARE INVESTED INTO
GROUND BREAKING MATHEMATICAL RESEARCH!

ONLY €50 PER TICKET

***PAYOUT SCHEDULE:**

WEEK 1:	€1
WEEK 2:	€1/2
WEEK 3:	€1/3
WEEK 4:	€1/4
WEEK 5:	€1/5
WEEK 6:	€1/6
...	...
WEEK N:	€1/N

EVERY TICKET
IS A WINNER!!!

IDEA #7: HOW TO FUND RESEARCH IN MATHEMATICS.

WE WON €750 TRILLION
EACH IN THE LOTTERY!!



HERE IS €10 EACH. THAT
SHOULD LAST YOU FOR
A LIFETIME OR TWO.



spikedmath.com
© 2012

KOMMENTAR: Divergenz ist nicht unbedingt eine schlechte Eigenschaft einer Folge. Z. B. wirst du später im Studium sehen, dass die Divergenz dieser Folge h , welche du später als harmonischen Reihe kennenlernen wirst, eine Verschärfung der Aussage ist, dass es unendlich viele Primzahlen gibt. (Stichwort: Riemannsche Zeta-Funktion)

Wir wollen unsere beiden Folgen von früher wieder aufgreifen:

> $a := n \rightarrow (1 + 1/n)^n$;

b:=n->(1+1/n)^(n+1);

$$a:=n\rightarrow\left(1+\frac{1}{n}\right)^n$$

$$b:=n\rightarrow\left(1+\frac{1}{n}\right)^{n+1} \quad (1.2.9)$$

> map(i->evalf(b(i)),[1..10]);

[4., 3.375000000, 3.160493827, 3.051757812, 2.985984000,
2.941897434, 2.910285368, 2.886507578, 2.867971991,
2.853116706] (1.2.10)

MATH: Die Folge b_n ist durch b_1 beschränkt, da sie (streng) monoton fällt und stets positiv ist. Damit ist sie konvergent.

ÜBUNG [03]:

1) Zeige, dass die Folge a_n konvergiert.

2) Zeige: $b_n > a_n > 0$ für alle $n \in \mathbb{N}$.

3) Zeige: Die Folge $n \mapsto \frac{b_n}{a_n}$ konvergiert gegen 1.

4) Zeige nun allgemein: Sind a_n und b_n Folgen mit $\frac{b_n}{a_n} \rightarrow 1, n \rightarrow \infty$, und ist a_n

beschränkt, dann ist $n \mapsto b_n - a_n$ eine Nullfolge. Folgere, dass die beiden Folgen denselben Grenzwert haben, falls dieser existiert.

MATH: Der (gemeinsame) Grenzwert der beiden Folgen a_n & b_n heißt Eulersche Zahl e .

Grenzwertsätze

MATH: Die reellwertigen Folgen bilden einen kommutativen Ring mit 1 bezüglich der wertweisen Addition und Multiplikation.

$$(a+b)_n := a_n + b_n$$

$$(a \cdot b)_n := a_n b_n$$

MATH: Wir wollen dies nicht formal beweisen, sondern begnügen uns mit der Feststellung, dass alle Ringaxiome auf ihre Gültigkeit im Wertebereich \mathbb{R} zurückgeführt werden. Die konstante Folge $n \mapsto 0$ und die konstante Folge $n \mapsto 1$ bilden jedenfalls das Null- und das Einselement.

MATH: Die konvergenten Folgen bilden einen Teilring vom Ring aller Folgen und die Limesbildung ist vertauschbar mit den Ringoperationen: Sind $(a_n), (b_n)$ reellwertige **konvergente** Folgen, so gilt:

$$\lim_{n \rightarrow \infty} (a_n + b_n) = \lim_{n \rightarrow \infty} a_n + \lim_{n \rightarrow \infty} b_n$$

$$\lim_{n \rightarrow \infty} (a_n \cdot b_n) = \lim_{n \rightarrow \infty} a_n \cdot \lim_{n \rightarrow \infty} b_n$$

DENKANSTOSS: Finde ein Beispiel, welches zeigt, dass Konvergenz nötig für die letzte Aussage ist.

ÜBUNG [04]:

Die Definition der Konvergenz forderte die Existenz eines $N_0(\square)$ mit gewissen Eigenschaften. Wir schreiben $N_0(a, \varepsilon)$, wenn wir uns auf die Folge a beziehen.

1) Drücke $N_0(a + b, \varepsilon)$ durch $N_0(a, f_1(\varepsilon))$ und $N_0(b, f_2(\varepsilon))$ für geeignete Funktionen f_1, f_2 aus.

2) Drücke $N_0(a \cdot b, \varepsilon)$ durch $N_0(a, g_1(\varepsilon))$ und $N_0(b, g_2(\varepsilon))$ für geeignete Funktionen g_1, g_2 aus.

MATH: Eine Sonderrolle nimmt die Division ein, die ja in einem Ring nicht uneingeschränkt ausführbar ist. Hier brauchen wir eine Zusatzvoraussetzung:

$b_n \neq 0$ für alle $n \in \mathbb{N}$ und $\lim_{n \rightarrow \infty} b_n \neq 0$. Dann ist $\frac{a_n}{b_n}$ auch konvergent mit

$$\lim_{n \rightarrow \infty} \left(\frac{a_n}{b_n} \right) = \frac{\lim_{n \rightarrow \infty} a_n}{\lim_{n \rightarrow \infty} b_n}.$$

Der Fall der Division lässt noch etwas Luft zum Experimentieren:

> **a:=1/(n^2+2);**
b:=1/(n^2-2);

$$a := \frac{1}{n^2 + 2}$$

$$b := \frac{1}{n^2 - 2}$$

(1.3.1)

> **limit(a, n=infinity);**
limit(b, n=infinity);
limit(a/b, n=infinity);

0

0

1

(1.3.2)

> **b:=1/n;**
limit(b, n=infinity);

$$b := \frac{1}{n}$$

0

(1.3.3)


```
> limit(a/b,n=infinity);
```

$$0 \quad (1.3.4)$$

```
> limit(b/a,n=infinity);
```

$$\infty \quad (1.3.5)$$

```
> b:=1/(-n)^n;
   limit(b,n=infinity);
```

$$b := \frac{1}{(-n)^n}$$
$$0 \quad (1.3.6)$$

```
> limit(b/a,n=infinity);
```

$$0 \quad (1.3.7)$$

```
> limit(a/b,n=infinity);
```

$$\infty \quad (1.3.8)$$

Würdest du diesem Ergebnis trauen? Nach allgemeinem Verständnis sagt man bei einer reellwertigen Folge $(c_n) : \mathbb{N} \rightarrow \mathbb{R}$

$$\lim_{n \rightarrow \infty} c_n = \infty,$$

wenn gilt: Zu jedem $k > 0$ existiert ein $n \in \mathbb{N}$ mit $c_i > k$ für alle $i > n$. Man sagt auch, dass die Folge **bestimmt divergiert** oder **gegen unendlich konvergiert**.

▼ Häufungspunkte von Folgen

MATH - BEISPIEL: Ist eine Folge (a_n) konvergent, so gilt dies auch für jede Teilfolge. Im Falle der Nichtkonvergenz kann es konvergente Teilfolgen geben:

```
> a:=3+1/n+(-1)^n;
```

$$a := 3 + \frac{1}{n} + (-1)^n \quad (1.4.1)$$

```
> map(i->evalf(subs(n=i,a)),[1..20]);
```

$$[3., 4.500000000, 2.333333333, 4.250000000, 2.200000000, \\ 4.166666667, 2.142857143, 4.125000000, 2.111111111, \\ 4.100000000, 2.090909091, 4.083333333, 2.076923077, \\ 4.071428571, 2.066666667, 4.062500000, 2.058823529, \\ 4.055555556, 2.052631579, 4.050000000]$$
$$(1.4.2)$$

Um zu sehen, dass die reelle Folge nicht konvergent ist, kann man sich mit einer graphischen Veranschaulichung helfen. Wir betrachten den Graphen der Folge $a : \mathbb{N} \rightarrow \mathbb{R}$ (also einer reellwertigen Funktion mit Definitionsbereich gleich den natürlichen Zahlen).

Dazu definieren wir zunächst die Folge als eine Abbildung:

```
> a:=n->3+1/n+(-1)^n;
```

$$(1.4.3)$$


$$a := n \rightarrow 3 + \frac{1}{n} + (-1)^n \quad (1.4.3)$$

Wir wollen nun die ersten 50 Elemente der Folge finden und in einer Liste von Punkten $[n, a_n]$ schreiben:

```
> A20:=map(n->[n, a(n)], [$1..50]):
```

Nun visualisieren wir die Elemente der Folge:

```
> plot(A20, view=[0..50,0..5], style=point);
```



Wir haben jetzt schon die Vermutung, dass die Folge nicht konvergent ist und lassen wir uns von der Visualisierung helfen, unsere Vermutung zu beweisen. Der Graph selbst ist jedoch noch kein Beweis. Wir betrachten also eine Teilfolge

$$ta1: \mathbb{N} \rightarrow \mathbb{R} : n \mapsto a_{2n}$$

(beachte die Definition von a)

```
> ta1:=a(2*n);
```

$$ta1 := 3 + \frac{1}{2n} + (-1)^{2n} \quad (1.4.4)$$

```
> map(i->evalf(subs(n=i,ta1)), [$1..20]);
```

(1.4.5)

```
[4.500000000, 4.250000000, 4.166666667, 4.125000000, 4.100000000, (1.4.5)
  4.083333333, 4.071428571, 4.062500000, 4.055555556,
  4.050000000, 4.045454545, 4.041666667, 4.038461538,
  4.035714286, 4.033333333, 4.031250000, 4.029411765,
  4.027777778, 4.026315789, 4.025000000]
```

Jetzt können wir (mit den Mitteln der Analysis) auch beweisen, dass die Folge konvergent ist gegen den Grenzwert 4.

Die folgende Teilfolge ist konvergent gegen den Grenzwert 2.

```
> ta2:=a(2*n+1);
```

$$ta2 := 3 + \frac{1}{2n+1} + (-1)^{2n+1} \quad (1.4.6)$$

```
> map(i->evalf(subs(n=i,ta2)),[$1..20]);
```

```
[2.333333333, 2.200000000, 2.142857143, 2.111111111, 2.090909091, (1.4.7)
  2.076923077, 2.066666667, 2.058823529, 2.052631579,
  2.047619048, 2.043478261, 2.040000000, 2.037037037,
  2.034482759, 2.032258065, 2.030303030, 2.028571429,
  2.027027027, 2.025641026, 2.024390244]
```

MATH: $h \in \mathbb{R}$ heißt **Häufungspunkt** (oder **Berührungspunkt**) der Folge $a = (a_n) \in \mathbb{R}^{\mathbb{N}}$, falls für jedes $m \in \mathbb{N}$ und jedes $\square > 0$ ein $n > m$ existiert mit $|a_n - h| < \square$.

Offenbar ist diese Bedingung äquivalent zu der Existenz einer Teilfolge mit Grenzwert h .

DENKANSTOSS: Beweise diese Äquivalenz.

MATH: Der grundlegende Existenzsatz für Häufungspunkte ist der folgende Satz von Bolzano-Weierstraß:

Jede beschränkte reellwertige Folge hat mindestens einen Häufungspunkt.

```
> a:=n->((-1)^n*n^5-n+(-1)^n)/(n^5-2);
```

$$a := n \rightarrow \frac{(-1)^n n^5 - n + (-1)^n}{n^5 - 2} \quad (1.4.8)$$

```
> evalf(map(a,[$2..20]));
```

```
[1.033333333, -1.024896266, 0.9990215264, -1.002561639, (1.4.9)
  0.9996140983, -1.000595061, 0.9998474028, -1.000203228,
  0.9999299986, -1.000086930, 0.9999638307, -1.000043093,
  0.9999795471, -1.000023704, 0.9999876022, -1.000014086,
  0.9999920617, -1.000008885, 0.9999946875]
```

Also sieht diese Folge beschränkt aus. Wie beweist man das?

Neben einer direkten Abschätzung des Bruchs nach oben und unten kann hierzu alternativ benutzt werden, dass jede konvergente Folge beschränkt ist (**DENKANSTOSS:** Warum?).

Die fragliche Folge kann nun in zwei Teilfolgen "zerlegt" werden, die beide konvergent, also insbesondere beschränkt sind. Damit können wir auf die Beschränktheit der gesamten Folge schließen. Für die Teilfolgen berechnen wir:

$$\text{> limit(simplify(a(2*n)),n=infinity);}$$

$$-1 - 1..1 + 1 \quad (1.4.10)$$

$$\text{> limit(simplify(a(2*n+1)),n=infinity);}$$

$$-1 - 1..1 + 1 \quad (1.4.11)$$

Da Maple nichts davon weiß, dass n eine natürliche Zahl ist, ist die Verwirrung von Maple nachvollziehbar. Was können wir tun?

$$\text{> simplify(a(2*n)) assuming n::positiv;}$$

$$\text{limit(%,n=infinity);}$$

$$1 \quad (1.4.12)$$

$$\text{> simplify(a(2*n+1)) assuming n::positiv;}$$

$$\text{limit(%,n=infinity);}$$

$$-1 \quad (1.4.13)$$

ÜBUNG [05]:

Gib eine Folge an, die unbeschränkt ist und genau zwei Häufungspunkte hat.

Das nächste Beispiel soll zeigen, dass es Folgen mit unendlich vielen Häufungspunkten gibt.

$$\text{> b := n -> n / 10^(1+floor(log[10](n)));}$$

$$b := n \rightarrow \frac{n}{10^{1 + \text{floor}(\log_{10}(n))}} \quad (1.4.14)$$

Wir betrachten die ersten Folgenglieder:


$$\text{> map(n -> b(n) , [1..10]);}$$

$$\left[\frac{1}{10}, \frac{1}{5}, \frac{3}{10}, \frac{2}{5}, \frac{1}{2}, \frac{3}{5}, \frac{7}{10}, \frac{4}{5}, \frac{9}{10}, \frac{1}{10} \right] \quad (1.4.15)$$

Wir visualisieren das Verhalten der Folge. Du musst hierbei nicht den Code verstehen, der die Animation erzeugt. Klicke auf das Bild, um oben die Option zum Abspielen zu erhalten.


$$\text{> with(plots);}$$

$$\text{> P := map(i -> pointplot(map(n -> [n,b(n)] , [1..(20*i+1)]) , color=[black$(20*i),red] , [0..15]); display(P, insequence=true);}$$



Besser ist die Animation, wenn wir die Folge nicht als Abbildung visualisieren, sondern nur die Punkte im Einheitsintervall einzeichnen. Beachte, dass der aktuelle Punkt rot ist.

```
> P2 := map( i -> pointplot( map( n -> [b(n),0] , [1..(10i+1)] ) , color=[black$(10i),red] , view=[0..1,-1/10,1/10], scaling=constrained) , [0..10] ):display(P2, insequence=true);
```



ÜBUNG [06]:

a) Zeige, dass alle Glieder der Folge b_n Häufungspunkte sind. b) Zeige, dass alle Zahlen im Intervall $(0.1, 1)$ Häufungspunkte der Folge b_n sind.

KOMMENTAR: Wenn du in den nächsten Wochen Abzählbarkeit kennen lernst, so erinnere dich an diese Folge und erkenne die Tatsache, dass sie überabzählbar viele Häufungspunkte hat.

▼ Limes superior und Limes inferior

MATH: Besonders wichtige Häufungspunkte von reellwertigen Folgen sind der **Limes superior** und der **Limes inferior**, soweit endlich. Ist (a_n) eine reellwertige Folge, so sei A die Menge ihrer Häufungspunkte.

$\limsup a_n := \sup(A)$, falls (a_n) nach oben beschränkt und A nicht leer.

$\limsup a_n := \infty$, falls (a_n) nicht nach oben beschränkt.

$\limsup a_n := -\infty$, falls (a_n) nach oben beschränkt und A leer.

Die Definition von \liminf ist entsprechend.

```
> a:=(-1)^n*n;
                                     a:= (-1)^n n
(1.5.1)
```

```
> limit(a,n=infinity);
                                     undefined
(1.5.2)
```

DENKANSTOSS: Man zeige für die gerade definierte Folge:

1.) $\limsup a_n = \infty$,

2.) $\liminf a_n = -\infty$,

und die Folge hat keine Häufungspunkte. (Hinweis: Teilfolgen.)

ÜBUNG [07]:

Zeige, dass sich die Grenzwertsätze nicht auf \limsup und \liminf übertragen lassen.

Hinweis:

```
> a:=(-1)^n;
  limit(a-a,n=infinity);
                                     a:= (-1)^n
                                     0
(1.5.3)
```

DENKANSTOSS: Falls $\liminf a_n$ endlich ist, so ist es ein Häufungspunkt der Folge.